

微型计算机BASIC算法语言

通用程序选介

刘 乐 融

(中国科学院西北水土保持研究所)

引进计算机, 外商往往在应用程序上“卡”我们; 花高价买来, 只能得到输出结果。刘乐融同志发愤学习, 编出一些通用程序, 在水土保持科学试验中有使用价值。本刊陆续选介其中几个, 供普及计算机、迎接新技术革命挑战的参考。——编者

目前, 我国现有的微型计算机存在着一个问题, 就是本机所带的应用程序很少或者就没有。即使从国外购进的微型计算机, 也因外商对应用程序采取保密措施, 使用者看不到程序清单。这对普及、推广微机的应用是很不利的。

用计算机算法语言进行程序设计, 既要了解数值计算方法, 又要熟悉计算机算法语言, 这需要花费很多劳动。为了避免不必要的重复劳动, 针对我们水土保持研究所科学研究工作的共同点, 先后完成了“单因素方差分析”、“多因素方差分析”、“多元线性回归分析”、“逐步回归分析”和“线性规划问题的解法”等应用程序的设计, 并全部在TRS-80 I 微机上通过。

本文对每个程序所用数学方法简单给予描述, 着重介绍各程序的使用及说明, 给出各程序的清单, 并附有例题。使用者只要给出具体数据, 按照例题的样式, 即可上机运算。

I ——多元线性回归分析程序

多元线性回归分析原理与一元线性回归分析完全相同, 只是在计算上要比一元线性回归分析复杂的多。一般要用电子计算机进行计算。

一、方法概要

设有自变量 X_1, X_2, \dots, X_m , 因变量 Y , 共作 n 次试验。若 Y 与 X_i ($i = 1, 2, \dots, m$) 间有线性关系, 回归方程则为:

$$\hat{Y} = b_0 + b_1 X_1 + b_2 X_2 + \dots + b_m X_m \quad (1)$$

显而易见, 只要确定了 $b_0, b_1, b_2, \dots, b_m$ 各回归系数, 方程(1)也就确定了。

若我们把得到的试验数据用矩阵的形式表示为:

$$X = \begin{bmatrix} X_{11} & X_{12} & X_{13} & \dots & X_{1n} \\ X_{21} & X_{22} & X_{23} & \dots & X_{2n} \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ X_{m1} & X_{m2} & X_{m3} & \dots & X_{mn} \end{bmatrix} \quad Y = \begin{bmatrix} Y_1 \\ Y_2 \\ \vdots \\ Y_m \end{bmatrix}$$

由最小二乘法知道, $b_0, b_1, b_2, \dots, b_m$ 应使得全部观察值 Y_m 与回归值 \hat{Y}_m 的偏差平方和 Q 达

到最小, 即使

$$Q = \sum_m (Y_m - \hat{Y}_m)^2$$

$$= \sum_m (Y_m - b_0 - b_1 X_{m1} - b_2 X_{m2} - \cdots - b_n X_{mn})^2 = \text{最小}$$

根据微积分学中的极值原理, $b_0, b_1, b_2, \cdots, b_n$, 应是下列方程组的解:

$$\frac{\partial}{\partial b_0} Q = -2 \sum_m (Y_m - \hat{Y}_m) = 0$$

$$\frac{\partial}{\partial b_j} Q = -2 \sum_m (Y_m - \hat{Y}_m) X_{mj} = 0 \quad (j = 1, 2, \cdots, n)$$

(2)

或

$$\sum_m (Y_m - b_0 - b_1 X_{m1} - b_2 X_{m2} - \cdots - b_n X_{mn}) = 0$$

$$\sum_m (Y_m - b_0 - b_1 X_{m1} - b_2 X_{m2} - \cdots - b_n X_{mn}) X_{mj} = 0$$

(3)

($j = 1, 2, 3, \cdots, n$)

显然方程组 (2) 的系数矩阵是对称矩阵, 若用 **A** 来表示它, 则 **A** = **XX'** 即:

$$\mathbf{A} = \begin{bmatrix} N & \sum_m X_{m1} & \sum_m X_{m2} & \cdots & \sum_m X_{mn} \\ \sum_m X_{m1} & \sum_m X_{m1}^2 & \sum_m X_{m1} X_{m2} & \cdots & \sum_m X_{m1} X_{mn} \\ \sum_m X_{m2} & \sum_m X_{m1} X_{m2} & \sum_m X_{m2}^2 & \cdots & \sum_m X_{m2} X_{mn} \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ \sum_m X_{mn} & \sum_m X_{m1} X_{mn} & \sum_m X_{m2} X_{mn} & \cdots & \sum_m X_{mn}^2 \end{bmatrix}$$

$$= \begin{bmatrix} 1 & 1 & \cdots & 1 & 1 & X_{11} & X_{12} & \cdots & X_{1n} \\ X_{11} & X_{21} & \cdots & X_{m1} & 1 & X_{21} & X_{22} & \cdots & X_{2n} \\ X_{12} & X_{22} & \cdots & X_{m2} & 1 & X_{31} & X_{32} & \cdots & X_{3n} \\ \vdots & \vdots & \ddots & \vdots & \vdots & \vdots & \vdots & \ddots & \vdots \\ X_{1n} & X_{2n} & \cdots & X_{mn} & 1 & X_{n1} & X_{n2} & \cdots & X_{nn} \end{bmatrix}$$

$$= \mathbf{X}' \mathbf{X}$$

方程组 (2) 右端常数项矩阵 **B** 也可用矩阵 **X** 和 **Y** 来表示:

$$\mathbf{B} = \begin{bmatrix} \sum_m Y_m \\ \sum_m X_{m1} Y_m \\ \sum_m X_{m2} Y_m \\ \vdots \\ \sum_m X_{mn} Y_m \end{bmatrix} = \begin{bmatrix} 1 & 1 & \cdots & 1 & Y_1 \\ X_{11} & X_{21} & \cdots & X_{m1} & Y_2 \\ X_{12} & X_{22} & \cdots & X_{m2} & Y_3 \\ \vdots & \vdots & & \vdots & \vdots \\ X_{1n} & X_{2n} & \cdots & X_{mn} & Y_m \end{bmatrix} = \mathbf{X}' \mathbf{Y}$$

这样方程组 (2) 的矩阵形式是

$$(\mathbf{X}' \mathbf{X}) \mathbf{b} = \mathbf{X}' \mathbf{Y}$$

或

$$\mathbf{A} \mathbf{b} = \mathbf{B}$$

那么就有: $b = A^{-1}B$

若设

$$A^{-1} = C = \begin{bmatrix} C_{00} & C_{01} & C_{02} & \dots & C_{0n} \\ C_{10} & C_{11} & C_{12} & \dots & C_{1n} \\ C_{20} & C_{21} & C_{22} & \dots & C_{2n} \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ C_{m0} & C_{m1} & C_{m2} & \dots & C_{mn} \end{bmatrix} \quad B = \begin{bmatrix} B_0 \\ B_1 \\ B_2 \\ \vdots \\ B_n \end{bmatrix}$$

就可以得到各回归系数

$$b_k = C_{k0}B_0 + C_{k1}B_1 + C_{k2}B_2 + \dots + C_{kn}B_n$$

$$k = 0, 1, 2, \dots, n$$

二、程序清单

```

1000 CLS
1010 INPUT "ROW :N1 AND COL :N2"; N1,N2 :
      LPRINT TAB (5) "N1="N1, "N2="N2
1020 DIM A (N1, N2), A1 (N2, N2), W (N2), B (N2),
      C (N2), P (N2), Q (N2), X1 (N2)
1030 LPRINT".....DATA....."
1040 FOR X=1 TO N1
1045 PRINT X, :LPRINT TAB (3) ; X,
1050 FOR Y=1 TO N2
1060 READ A (X, Y)
1070 PRINT A (X, Y) ; :LPRINT TAB (Y*10)
      A (X, Y) ;
1080 NEXT Y
1090 PRINT"  " :LPRINT "  "
1100 NEXT X :LPRINT "  " :LPRINT "  "
1110 A1 (1, 1) = N1
1120 FOR Y=1 TO N2-1
1130 L=0
1140 FOR X=1 TO N1
1150 L=L+A (X, Y)
1160 NEXT X
1170 I=I+1 :J=I
1180 A1 (J+1, 1) =L :A1 (1, J+1) =L
1190 NEXT Y
1200 C=C+1 :W=W+1
1210 FOR Y=1 TO N2-1
1220 L=0
1230 FOR X=1 TO N1
1240 L=L+A (X, Y) * A (X, C)
1250 NEXT X
1260 A1 (Y+1, W+1) =L
1270 NEXT Y

```

```

1280 IF C<>N2-1 THEN 1200
1290 L=O:Y1=O:C1=O
1300 FOR X=1 TO N1
1310 L=L+A (X, N2) : Y1=Y1+A (X, N2) ^2
1320 NEXT X
1330 W (1) =L:C1=W (1) ^2
1340 FOR Y=1 TO N2-1
1350 L=O
1360 FOR X=1 TO N1
1370 L=L+A (X, N2) * A (X, Y)
1380 NEXT X
1390 W (Y+1) =L
1400 NEXT Y
1410 FOR K=1 TO N2
1420 W=O
1430 FOR I=K TO N2
1440 FOR J=K TO N2
1450 IF ABS (A1 (I, J) ) >ABS (W) THEN W=A1 (J, J)
      : IO=I:JO=J
1460 NEXT J
1465 NEXT I
1470 IF ABS (W) <=10\ (-12) THEN PRINT "FAIL"
1480 IF IO=K THEN 1520
1490 FOR J=1 TO N2
1500 Z=A1 (IO, J) : A1 (IO, J) =A1 (K, J) :
      A1 (K, J) =Z
1510 NEXT J
1520 IF JO=K THEN 1560
1530 FOR I=1 TO N2
1540 Z=A1 (I, JO) : A1 (I, JO) =A1 (I, K) :
      A1 (I, K) =Z
1550 NEXT I
1560 P (K) =IO:Q (K) =JO
1570 FOR J=1 TO N2
1580 IF J=K THEN B(J) =1/W:C (J) =1 ELSE
      B (J) =-A1 (K, J) /W:C (J) =A1 (J, K)
1590 A1 (J, K) =O:A1 (K, J) =A1 (J, K)
1600 NEXT J
1610 FOR I=1 TO N2
1620 FOR J=1 TO N2
1630 A1 (I, J) =A1 (I, J) +C (I) * B (J)
1640 NEXT J
1645 NEXT I
1650 NEXT K

```

```

1660 FOR K=N2 TO 1 STEP -1
1670 IO=P (K) : JO=Q (K)
1680 IF IO=K 1720
1690 FOR I=1 TO N2
1700 Z=A1 (I, IO) : A1 (I, IO) =A1 (I, K) :
      A1 (I, K) =Z
1710 NEXT I
1720 IF JO=K 1760
1730 FOR J=1 TO N2
1740 Z=A1 (JO, J) : A1 (JO, J) =A1 (K, J) :
      A1 (K, J) =Z
1750 NEXT J
1760 NEXT K
1770 FOR X=1 TO N2
1780 L=O
1790 FOR Y=1 TO N2
1800 L=L+A1 (X, Y) * W (Y)
1810 NEXT Y
1820 X1 (X) =L
1830 PRINT "B"X-1"="; L : LPRINT "B"X-1"="; L
1840 NEXT X
1850 LPRINT " " : LPRINT " " : L=O
1860 FOR X=1 TO N2
1870 L=L+X1 (X) * W (X)
1880 NEXT X
1890 S=O : S1=O : S=Y1-L : S1=Y1- (1/N1) *C1
1900 PRINT "R="; (1-S/S1) ^ (1/2) , "SURPLUS="; S,
      "SUM="; S1, "SREG="; S1-S
1910 LPRINT "R="; (1-S/S1) ^ (1/2) , "SURPLUS="; S,
      "SUM="; S1, "SREG="; S1-S : LPRINT " "
1920 FO= ( (S1-S) / (N2-1) ) / (S/ (N1-N2) )
1930 PRINT "FO="; FO : LPRINT "FO="; FO : LPRINT
1940 FOR Y=2 TO N2
1950 P=X1 (Y) ^2/A1 (Y, Y) : T=P/ (1/2) / (S/ (N1-N2) )
      ^ (1/2) : F=P/ (S/ (N1-N2) )
1960 PRINT "P"Y-1"="P, TAB (25) "T"Y-1"="T, TAB (50)
      "F"Y-1"="F : LPRINT "P"Y-1"="P, TAB (25)
      "T"Y-1"="T, TAB (50) "F"Y-1"="F
1970 NEXT Y
1980 RETURN

```

三、使用说明

N_1 —试验次数

N_2 —自变量总个数 + 1

B_i —各回归系数。 ($i = 0, 1, 2, 3 \dots$)

SURPLUS—剩余平方和。

SPEG—回归平方和。

SUM—总计平方和。

R—复相关系数。

FO—F比。

Ti—各因子的t检验。 ($i = 1, 2, \dots$)

Fi—各因子的F检验。 ($i = 1, 2, 3 \dots$)

P—偏回归平方和。

例题：研究同一地区土壤内所含植物可给态磷的情况，得到18组数据如下，其中

X_1 —土壤内所含无机磷浓度；

X_2 —土壤内溶于 K_2CO_3 溶液并受溴化物水解的有机磷；

X_3 —土壤内溶于 K_2CO_3 溶液但不溶于溴化物的有机磷；

y—栽在 $2^\circ C$ 土壤内的玉米中的可给态磷。已知y对 X_1, X_2, X_3 存在着线性相关，求它们的回归方程，并进行检验。

土壤样本	X_1	X_2	X_3	Y
1	0.4	53	158	64
2	0.4	23	163	60
3	3.1	19	37	71
4	0.6	34	157	61
5	4.7	21	59	54
6	1.7	65	123	77
7	9.4	14	46	81
8	10.1	31	117	93
9	11.6	29	173	93
10	12.6	58	112	51
11	10.9	37	111	76
12	23.1	46	114	96
13	23.1	50	134	77
14	21.6	44	73	93
15	23.1	56	168	95
16	1.9	36	143	54
17	26.8	58	202	168
18	29.9	51	124	99

100 DATA 0.4, 53, 158, 64, 0.4, 23, 163, 60
 110 DATA 3.1, 19, 37, 71, 0.6, 34, 157, 61
 120 DATA 4.7, 24, 59, 54, 1.7, 65, 123, 77
 130 DATA 9.4, 44, 46, 81, 10.1, 31, 117, 93
 140 DATA 11.6, 29, 173, 93, 12.6, 58, 112, 51
 150 DATA 10.9, 37, 111, 76, 23.1, 46, 114, 96

```

160 DATA 23.1, 50, 134, 77, 21.6, 44, 73, 93
170 DATA 23.1, 56, 168, 95, 1.9, 36, 143, 54
180 DATA 26.8, 58, 202, 168, 19.9, 51, 124, 99
190 GOSUB 1000
200 END

```

键 入

显 示

RUN RETURN

N1 = ?

18 RETURN

N2 = ?

4 RETURN

N1 = 18 N2 = 4

.....DATA.....

NO.	X1	X2	X3	Y
1	0.4	53	158	64
2	0.4	23	163	60
3	3.1	19	37	71
4	0.6	34	157	61
5	4.7	24	59	54
6	1.7	65	123	77
7	9.4	44	46	81
8	10.1	31	117	93
9	11.6	29	173	93
10	12.6	58	112	51
11	10.9	37	111	76
12	23.1	46	114	96
13	23.1	50	134	77
14	21.6	44	73	93
15	23.1	56	168	95
16	1.9	36	143	54
17	26.8	58	202	168
18	29.9	51	124	99

B0 = 43.6522

B1 = 1.78478

B2 = -0.0833964

B3 = 0.161133

R = 0.741177 SURRLUS = 5583.41

SUM = 12389.5 SREG = 6806.08

FO = 5.68858

P1 = 4394.15 T1 = 3.31934 F1 = 11.018

P2 = 15.8977 T2 = 0.199655 F2 = 0.398622

P3 = 830.444 T3 = 1.44301 F3 = 2.08228

Y = B0 + B1X₁ + B2X₂ + B3X₃